
“나는아직내경력의시작단계에있다. 모든것이새롭고, 나는여전히가야하기에배운다.”

─Orlando Bloom

배경지식-옮긴이작성

3장에서의 다소 생소한 개념은 믹스인 클래스이다. 믹스인 클래스는 구현 부분인 몸체

를 가지는 인터페이스와 같은 개념으로 생각하거나, C/C++ 언어에서의 함수 모음인

C/CPP 파일로 보면 무난할 것으로 보인다. 즉, 공통적인 부분을 미리 구현해 놓은 후

상속하여 원하는 함수를 오버라이딩해서 사용하거나 원치 않는 부분은 오버라이딩하지

않아도 되는 점이 인터페이스나 추상 클래스와는 다른 점이다.

컴파일을 하면“클래스명$Mixin.class”형태의 함수가 생성되는 것도 다른 점이다. 다

시 말하자면, 일종의 함수 모음집으로 상속한 모든 믹스인 클래스의 함수와 요소들은 클

래스의 일부분으로 속하여 컴파일된다.

이 외에 객체 상수 개념은 자바스크립트에 익숙한 프로그래머라면 이미 사용하고 있는

방식이며, 이 단원에서는 JavaFX의 기본 구문에 대해 설명하고 있으므로 이미 알고 있

는 개념이라 하더라도 꼼꼼히 살펴보기 바라며, 특히 일반적인 프로그래밍 언어에서는

제공하지 않는 시퀀스 등의 개념을 살펴보는 것도 좋을 것이다.

JavaFX 스크립트 기초

JavaFX는 부분적으로 선언형 언어이다. 선언형 언어를 사용하여 개발자는 할일을 기술

하고 시스템이 처리하도록 알린다. Chalmers 대학교의 인터렉션 디자인 기술 마스터 프

로그램의프로그램이사와 Göteborg 대학교부교수를역임하고있는 Olof Torgersson는

10년 넘게 선언형 프로그래밍을 연구해오고 있다. 선언형 프로그래밍에 관한 그의 연구

결과에서우리는다음의정의를찾았다.

“프로그래머 관점에서 기본 속성은 프로그래밍을 보다 높은 레벨로 추상화하는 것이

다. 높은 레벨의 추상화에서 프로그래머는 어떻게 계산되는지 알 필요가 없고, 무엇을

계산할 것인지에만 집중할 수 있다.”역주 1)

JavaFX 스크립트에는 객체지향과 선언형 프로그래밍의 개념이 섞여 있다. 이것은 높은

생산성은물론유연하고튼튼한애플리케이션의기초를제공한다. 그러나이런유연함은

역주 1) Torgersson, Olof. “A Note on Declarative Programming Paradigms and the Future of Definitional Program-

ming,”Chalmers University of Technology and Göeborg University, Göeborg, Sweden. http://www.cs.

chalmers.se/~oloft/Papers/wm96/wm96.html.

개발자의책임이필요하다. JavaFX 스크립트는관대한언어이고선언형이기에알기어려

운 프로그래밍 실수를 만드는 태생적인 규칙을 포함한다. 대부분의 경우는 널 객체가 런

타임 엔진에서 처리되고 Java 널 포인터 예외처리가 거의 일어나지 않기 때문이다. 그 결

과 프로그램은 표현식 안에서 널을 만날 때까지 계속되고 유효한 결과를 만들어낸다. 그

러나이결과는개발자가기대한값이아닐수도있다. 그러므로개발자는코드를작성할

때 더 주의해야 하고 더 많이 테스트해야 한다. 처음에는 이런 점을 주의해야 하지만

JavaFX는 쉬운 사용법과 높은 생산성을 지니고 있고 사용자가 문제를 겪지 않도록 노력

하고있어이러한단점을충분히상쇄한다.

선언형 언어인 JavaFX의 이점 중 하나는 객체가 상호작용을 하도록 만드는 많은‘연동’

방법을언어차원에서지원한다는것이다. 따라서개발자는어떻게보여주는지에얽매이

지 않고 무엇을 보여줄지에 집중할 수 있다. 다음 절에서 구문, 연산자와 그 외의 특징들

을포함한 JavaFX 스크립트언어의전반적인개요를볼수있다.

JavaFX 스크립트 언어

이미언급했듯이 JavaFX 스크립트는객체지향을지원하는선언형스크립팅언어이다. 이

미 Java, JavaScript, Groovy, Adobe Action Script, JRuby 등의다른언어를잘알고있다

면, JavaFX 스크립트가 낯익겠지만 JavaFX에는 중요한 차이점이 있다. 전통적인 순수 스

크립팅을 지원하면서 또한 객체지향적인 캡슐화와 코드의 재사용성을 지원한다. 따라서

개발자가 작은 크기부터 큰 크기까지 애플리케이션을 만들고 유지보수할 수 있다. 다른

중요한특징은 JavaFX 스크립트와 Java의혼용이부드럽게이루어진다는것이다.

개념적으로 JavaFX 스크립트는스크립트와클래스라는두개의레벨로나뉜다. 스크립트

레벨에서는 변수와 함수를 정의한다. 이 변수와 함수들은 같은 스크립트 안에서 정의한

다른 클래스에서 공유할 수 있고, 만약 스크립트가 더 큰 접근 권한을 가지고 있다면 다

른 스크립트와 클래스에서도 공유할 수 있다. 추가적으로 loose라고 부르는 표현식이 만

들어지는데, 이는 클래스 정의 밖에서 선언한 모든 표현식들이다. 스크립트가 수행될 때

모든 loose 표현식도수행된다.

간단하게‘Hello World’를콘솔에나타내는스크립트는다음과같다.

println(”Hello World”);

리스트 3.1은 3팩토리얼(3!)을계산하는것을보여준다.

리스트 3.1 3팩토리얼

def START = 3;

var result = START;

var a = result – 1;

while(a > 0) {

result *= a;

a––;

}

println(”result = {result}”);

만약 스크립트가 멤버들(외부 접근이 가능한 package, protected, public 등의 함수나 변수 같은

멤버들)를 익스포트하 다면 모든 loose 표현식이 run 함수에 반드시 포함되어야 한다. 예를 들

어, 앞의 예제에서 public을 추가하기 위해 result 변수를 바꾼다면 run 함수를 만들어야 한다. 

public var result:Number;

function run(args : String[]) : java.lang.Object {

var num = if(sizeof args > 0) {

java.lang.Integer.valueOf(args[0]);

} else {

10;

};

result = num;

var a = result – 1;

while(a > 0) {

result *= a;

a––;

}

println(”{num}! = {result}”);

}

run 메소드는 실행될 때 커맨드라인에서 스크립트에 전달한 인수들을 String[] 시퀀스 인수로

받는다.

개발자 노트 DEVELOPER NOTE

개발자 노트 DEVELOPER NOTE

(계속)

멤버들을 익스포트하지 않더라도 run 메소드를 계속 포함할 수 있다. 그러나 run 메소드는 비록

접근권한자가포함되지않았더라도자체적으로익스포트된것으로간주한다. 그래서일단 run메

소드를추가하면모든익스포트된 loose 표현식은반드시 run 메소드안에포함되어야한다.

스크립트 레벨과 별개로, 클래스는 인스턴스 변수와 함수를 정의하고사용하기 전에 반

드시 객체로 인스턴스화해야 한다. 클래스 함수나 변수는 같은 스크립트 파일 안의 스크

립트 레벨 함수와 변수에 접근할 수 있고, 접근 권한을 받은 경우에는 다른 스크립트 파

일안의함수와변수도접근가능하다. 반면에스크립트레벨함수는클래스가객체로생

성되고, 접근 권한을 제공하 을 때만 클래스 변수와 함수에만 접근할 수 있다. 접근 권

한에대한정의는이번장뒷부분에서자세히다룬다.

클래스 선언

JavaFX에서클래스를선언할때는 class키워드를사용한다. 

public class Title {

}

관습상 클래스명의 첫 자는 대문자로 쓴다.

접근 권한자 public키워드는 클래스가 다른 스크립트 파일에서 정의되었더라도 이 클래

스를 다른 클래스나 스크립트에서 사용될 수 있다는 것을 의미한다. 만약 클래스가 접근

권한자를 가지고 있지 않다면, 클래스가 선언된 스크립트 파일 안에서만 접근이 가능하

다. 예를 들어, 리스트 3.2에서 Point클래스는 외부에 보이는 권한자(public 같은)를 가지

고있지않으므로 ArtWork스크립트안에서만사용될수있다.

리스트 3.2 Artwork.fx

class Point { // private class only

// visible to the ArtWork class

var x:Number;

var y:Number;

}

public class ArtWork {

var location: Point;

}

각각의 JavaFX 스크립트 파일에서 비록 명백하게 정의하지 않더라도 스크립트 파일명을 사

용하여 클래스가 생성된다. 앞의 ArtWork.fx 예제에서도 Artwork란 클래스가 있다. Public

class Artwork 선언을 하지 않았어도 그렇게 됐을 것이다.

또한 같은 스크립트 파일 안에서 정의된 모든 클래스 이름에 스크립트 파일의 이름이 앞에 붙

는다. 예를 들어, 앞의 예제에서 Point 클래스는 ArtWork.Point와 같이 패키지 경로를 포함한

클래스 이름으로 간주된다. 물론, ArtWork가 패키지에 속한다면 패키지 이름 또한 이름으로 간

주될 것이다. 예를 들어 com.acme.ArtWork.Point가 되는 것이다.

클래스를 확장하려면 일반적인 클래스 이름 뒤에 extends라는 키워드를 사용하면 된다.

JavaFX 클래스는 하나의 Java나 JavaFX 클래스로 확장할 수 있다. Java 클래스를 확장하

면클래스는반드시디폴트(매개변수가없는) 생성자를갖는다.

public class Porsche911 extends Porsche {

}

JavaFX는다수의 JavaFX 믹스인클래스나 Java 인터페이스로확장한다. 믹스인클래스는

다음절에서다룬다.

애플리케이션은많은클래스를포함할수있어서 패키지Package라고하는일관된방식으로

클래스들을 구성하는 것이 좋다. 클래스나 스크립트가 패키지에 속하도록 선언하려면 스

크립트 파일의 시작부분에 패키지 선언을 포함한다. 다음 예제는 Title 클래스가

com.mycompany.components패키지에속한다는것을의미한다. 이제 Title클래스의완전한이

름은 com.mycompany.componets.Title이다. Title클래스가참조될때마다이완전한이름으로

풀어서사용한다.

package com.mycompany.components;

개발자 노트 DEVELOPER NOTE

public class Title {

}

이름을풀어쓰는것을쉽게하기위해 import 문을소스파일의윗부분에포함한다. 

import com.mycompany.components.Title;

var productTitle = Title{};

이제 Title이 스크립트 파일 안에서 참조되는 곳마다 com.mycompany.componetns.Title로

바뀌어사용된다. 또한와일드카드(*)를써서 import 선언을할수있다.

import com.mycompany.components.*;

와일드카드를포함한 import이기에 com.mycompany.components패키지에서어떤클래스를

참조하더라도 완전한 이름으로 바뀌어 쓰인다. 다음의 예제 코드는 주석에 완전한 클래

스이름을적어서클래스이름이어떻게풀어지는지보여준다.  

package com.mycompany.myapplication;

import com.mycompany.components.Title;

// com.mycompany.myapplication.MyClass

public class MyClass {

// com.mycompany.components.Title

public var title: Title;

}

클래스는 public 대신에 package 키워드를 사용하여 패키지 가시성을 가질 수 있다. 이것

은클래스가오직같은패키지안의클래스로부터접근될수있다는의미이다.

package class MyPackageClass {

}

클래스는 또한 abstract으로 선언될 수 있는데, 이것은 이 클래스가 바로 인스턴스화될 수

없다는 것을 의미한다. 그러나 서브클래스들 중 하나를 사용하여 인스턴스화될 수 있다.

추상클래스는독립적으로쓰일수없지만여러클래스에서사용하도록상태정보와함수

의 일부분을 캡슐화한다. 추상 클래스의 서브클래스만이 인스턴스화될 수 있고, 일반적으

로서브클래스는추상화클래스에서다루지않는고유의정보와기능을담아야한다. 

public abstract class MyAbstractClass {

}

클래스가 abstract 함수를선언한다면이클래스는 abstract으로선언되어야한다.

public abstract class AnotherAbstractClass {

public abstract function

setXY(x:Number, y:Number) : Void;

}

믹스인 클래스

JavaFX는 믹스인 상속이라는 상속의 형태를 지원한다. 이를 위해 JavaFX는 mixin이라는

특수한 타입의 클래스를 포함한다. mixin 클래스는 서브클래스에 상속시키는 특정 기능

을 제공하는 클래스이다. 이 클래스는 스스로 인스턴스화될 수 없다. mixin 클래스는 함

수들의 기본 구현을 제공하고, 변수를 선언하고 초기화한다는 점에서 Java 인터페이스와

다르다.

JavaFX에서 mixin 클래스를선언하려면클래스선언에 mixin키워드를포함해야한다. 

public mixin class Positioner {

mixin 클래스는 많은 함수 선언을 포함할 수 있다. 함수 선언이 함수 몸체를 가지고 있으

면 그것이 함수의 기본적인 구현이다. 예를 들어, 다음의 예제는 노드 안에 다른 노드가

들어있는 mixin 클래스선언이다.

public mixin class Positioner {

protected bound function centerX(node: Node, within: Node) : Number {

(within.layoutBounds.width – node.layoutBounds.width)/2.0 – node.layoutBounds.minX;

}

protected bound function centerY(node: Node, within: Node) : Number {

(within.layoutBounds.height – node.layoutBounds.height)/2.0 – node.layoutBounds.minY;

}

}

고유 버전의 mixin 함수를 서브클래스에 구현하려면 함수를 선언할 때 반드시 override키

워드를 사용해야 한다. 예를 들어, 다음의 코드는 Positioner mixin 클래스에서 centerx()란

함수의고유버전을구현한것이다.

public class My Positioner extends Positioner {

public override bound function centerX(node: Node, within: Node) : Number {

(within.boundsInParent.width – node.boundsInParent.width )/2.0;

}

}

mixin 함수가 기본적인 구현을 하지 않고 있다면, abstract으로 선언되어야 하고 서브 클

래스는 이 함수를 오버라이드해서 구현해야 한다. 예를 들어, 다음의 코드는 Positioner

mixin 클래스에추가한 abstract함수를보여준다.

public abstract function bottomY(node: Node, within: Node, padding: Number) : Number;

서브클래스는 다음의 코드에서 보여주듯이 override 키워드를 사용하여 이 함수를 구현

해야한다.

public class My Positioner extends Positioner {

public override function bottomY(node: Node, within: Node, padding: Number) : Number {

within.layoutBounds.height – padding – node.layoutBounds.height;

}

}

두 mixin이 같은 함수나 변수 이름을 가지고 있다면 시스템은 어떤 mixin이 extends절에

서 먼저 선언되었는지에 따라 함수나 변수를 사용한다. 특정 함수나 변수를 지정하려면

mixin 클래스이름을함수나변수이름과함께사용하면된다. 다음의코드를보자.

public class My Positioner extends Positioner, AnotherPositioner {

var offset = 10.0;

public override bound function

centerX(node: Node, within: Node) : Number {

Positioner.centerX(node, within) + offset;

}

}

mixin은 변수를 선언할 때 디폴트 값과 트리거를 가지게 할 수도 있고 안 가지게 할 수도

있다. 서브클래스는 이 변수를 상속하거나 변수 선언을 오버라이드해야 한다. 다음의 예

제를보자.

public mixin class Positioner {

public var offset: Number = 10.0;

}

public class My Positioner extends Positioner {

public override var offset = 5.0 on replace {

println(”{offset}”);

}

}

클래스가 JavaFX 클래스와 하나 이상의 mixin 클래스를 확장한다면, JavaFX 클래스는 변

수 초기화에서 mixin 클래스에 우선한다. 변수를 수퍼클래스에서 선언했다면 수퍼클래스

에서 지정한 디폴트 값을 사용한다. 디폴트 값을 수퍼클래스에서 지정하지 않았다면 변

수의 타입에 해당하는‘디폴트 값’이 사용된다. mixin 클래스에서 우선순위는 extends절

에서 정의한 순서를 따른다. mixin에서 선언된 변수가 디폴트 값을 가지고, 변수가 메인

클래스에서디폴트값없이오버라이드되었다면초기값은 mixin에서정한값으로지정된

다. 

mixin들은 또한 init와 postinit블록을 가지고 있다. mixin init와 postinit블록은 수퍼클

래스의 init와 postinit 블록 후에, 그리고 서브클래스의 init와 postinit 블록 전에 실행

된다. mixin 클래스의 init와 positnit블록은 서브클래스에서 extends절에서 선언된 순서

로실행된다.

객체 상수

JavFX에서 객체는 객체 상수object literals를 사용하여 인스턴스화한다. 객체 상수는 만들고

싶은클래스이름뒤에인스턴스에대한초기화와선언목록을적는선언적구문이다. 리

스트 3.3에서 Title 클래스의 객체는 스크린의 10, 50 위치에서“JavaFX is cool”이란 텍

스트를가지고만들어졌다. 마우스를클릭하면지정한함수를호출한다.

리스트 3.3 객체 상수

var title = Title {

text: ”JavaFX is cool”

x: 10

y: 50

onMouseClicked: function(e:MouseEvent):Void {

// 여기에 사용자 정의 코드를 작성한다.

}

};

객체상수를선언할때인스턴스변수는쉼표나공백, 세미콜론으로구분된다.

객체 상수 선언에서 추상함수를 오버라이드할 수 있다. 리스트 3.4에서 객체 상수는 java.

awt.event.ActionListener 인터페이스의 객체를 생성하고, 추상 Java 메소드인 void

actionPer-formed(ActionEvent e)메소드를오버라이드한다.

리스트 3.4 객체 상수-추상 함수 오버라이딩

import java.awt.event.ActionListener;

import java.awt.event.ActionEvent;

var listener = ActionListener {

override function actionPerformed(e: ActionEvent) : Void {

println(”Action Performed!”);

}

}

변수

JavaFX는 인스턴스instance와 스크립트script, 두 종류의 변수를 제공한다. 스크립트 변수는

스크립트 전체에 걸쳐 정보를 가지고, 인스턴스 변수는 스크립트 파일에서 선언한 클래

스의특정인스턴스에서만정보를가진다.

기본적으로변수는두종류이다. 바꿀 수 없는unassignable 것과바꿀 수 있는changeable 것. 바

꿀수없는변수는 def 키워드를사용해서선언하고한번값을정하면바꿀수없다.

public def PI = 3.14;

이 변수는 객체 상수에서 초기화, 오버라이드, 할당 모두 할 수 없다. 이런 점 때문에 이

변수를 상수로 볼 수 있지만 순수 상수는 아니고 바인딩을 할 수 있다(바인딩은 4장에서

자세히다룬다). 

다음 예제에서는 객체를 포함하는 바꿀 수 없는 변수를 선언한다. 객체 인스턴스는 변할

수없지만이인스턴스의상태도변할수없는것은아니다. 

def centerPoint = Point{x: 100, y:100};

centerPoint.x = 500;

실제로 centerPoint에할당한 point객체는변하지않고남아있다. 그러나 x와 y값인이객

체 인스턴스의 상태는 변할 수 있다. 그렇더라도 바인딩에 사용할 때의 centerPoint는 상

수다. 따라서 centerPoint의상태가변하면바인딩되어있으므로그변화를알수있다.

바꿀수있는인스턴스변수는 var키워드를사용하여선언하며디폴트값을지정할수있

다. 디폴트 값이 없으면 적당한 디폴트 값을 사용한다. 수는 0, Boolean 디폴트는 false,

String 디폴트는빈문자열, Sequence는빈시퀀스, 그리고나머지모두는 null로지정된다.

스크립트변수는클래스선언의외부에서선언하고인스턴스변수는클래스선언안에서

선언한다. 스크립트 변수를 접근 권한자 public, protected, package로 선언하면 스크립트

밖에서 사용할 때 완전한 이름으로 참조된다. 완전한 이름은 패키지 이름, 스크립트 이름,

변수 이름을 합친 것이다. 다음은 십자모양의 커서를 만들기 위해 javafx.scene.Cursor에

서사용한 public 스크립트변수의완전한이름이다.

javafx.scene.Cursor.CROSSHAIR;

인스턴스 변수는 클래스 선언부 안에서 선언되고 객체가 생성될 때 만들어진다. 리스트

3.5에서는스크립트변수와인스턴스변수의몇가지예를든다. 

리스트 3.5 스크립트 변수와 인스턴스 변수

import javafx.scene.Cursor;

import javafx.scene.paint.Color;

// javafx.scene.Cursor에서 스크립트 변수를 import

import javafx.scene.Cursor.CROSSHAIR;

// 바꿀 수 없는 스크립트 변수

def defaultText = ”Replace ME”; // 스크립트에서만 접근 가능

// 바꿀 수 있는 스크립트 변수

public var instanceCount: Integer; // Public 접근

public class Title {

// 바꿀 수 없는 인스턴스 변수

def defStroke = Color.NAVY; // 클래스에서만 접근

// 완전한 이름은 javafx.scene.paint.Color.NAVY

// 바꿀 수 있는 인스턴스 변수

// 디폴트 값은 빈 문자열인 ""

public var text:String;

public var width: Number; // 디폴트 값은 0.0

public var height = 100; // 정수 타입으로 예측

public var stroke: Color = defaultStroke;

public var strokeWidth = 1.0; // 숫자 타입으로 예측

public var cursor = CROSSHAIR;

// 완전한 이름은 javafx.scene.Cursor.CROSSHAIR

...

}

선언중어떤것들은타입을적고어떤것들은안적고있음을알수있다. 타입을선언하

지 않으면 처음 할당한 값에서 타입을 추론한다. String, Number, Integer 그리고 Boolean은

빌트-인타입이고나머지는 JavaFx나 Java 클래스이다. (Duration과 KeyFrame클래스인스턴스

를쉽게선언하는특수구문이있는데이것은 7장에서다룬다.)

표 3.1은 변수를 위한 접근 권한자의 목록과 각각의 의미 및 한계를 보여준다. 초기화할

때참조할수있으며, 객체상수선언시에도참조할수있다. 또한, 바인딩되는변수를알

수있다. 이것은 JavaFX의중요한특징이고 4장에서자세히다룰것이다.

또한 변수에 변환 트리거를 선언할 수 있다. 변환 트리거는 변수의 값이 변할 때마다 호

출되는 JavaFX 스크립트블록이다. 변환트리거는 on replace구문을통해선언한다.

public var x:Number = 100 on replace {

println(”New value is {x}”);

};

public var width: Number on replace (old) {

println(”Old value is {old}, New value is {x}”);

}

변환트리거는 4장에서더자세히다룬다.

시퀀스

시퀀스는객체를나열한리스트이다. 프로그래밍에서이런리스트가자주사용되기때문

에 JavaFX는 중요한 클래스 특징으로 시퀀스를 지원한다. 시퀀스를 선언하고, 아이템을

표 3.1 접근 권한자

접근 권한자 의미

var 디폴트 접근 허가는 스크립트 접근이고, 접근 권한자 없이 변수는 초기

화될 수 있다. 오버라이드, 읽기, 할당, 바인딩은 스크립트 안에서만 가

능하다.

def 디폴트 접근 허가는 스크립트 접근이다. 스크립트 안에서만 정의를 읽거

나 바인딩할 수 있다.

public var 누구나 읽고 쓸 수 있다. 또한 어디서든 초기화, 오버라이드, 읽기, 할

당, 바인딩이 가능하다.

public def 이 정의는 어디서든 읽을 수 있다. 접근 권한에 관계 없이 할당, 초기화

(객체 상수에서), 오버라이드가 불가능하다. 어디서든 바인딩이 가능하다.

public-read var 누구나 읽을 수 있지만 스크립트 안에서만 쓸 수 있다. 

public-init var 객체 상수에서 초기화될 수 있지만, 자신의 스크립트에서만 업데이트될

수 있다. 오직 인스턴스 변수만 가능하다.

package var 변수는 패키지에서 접근 가능하다. 이 변수는 같은 패키지 안의 클래스

에서 초기화, 오버라이드, 읽기, 할당, 바인딩이 가능하다.

package def 같은 패키지 안의 클래스에서 읽거나 바인딩할 수 있다.

protected var 패키지나 서브클래스에서 접근 가능하다. 같은 패키지의 클래스나 서브

클래스에서 초기화, 오버라이드, 읽기, 할당, 바인딩 가능하다.

protected def 같은 패키지의 클래스나 서브클래스에서 읽을 수 있고 바인딩 가능하다.

public-read protected var 누구나 읽고 바인딩할 수 있지만 이 변수는 같은 패키지의 클래스나 서

브 클래스에서만 초기화, 오버라이드, 할당이 가능하다.

public-init protected var 객체 상수에서 초기화할 수 있고, 누구나 읽고 바인딩할 수 있다. 그러

나 같은 패키지 안의 클래스나 서브클래스에서만 오버라이드나 할당이

가능하다. 오직 인스턴스 변수만이 가능하다.

추가하고, 아이템을 삭제하고, 아이템을 수정하는 동작들을 언어 자체에서 지원한다. 또

한시퀀스에서아이템을가져오는기능을강력히지원한다.

시퀀스선언

시퀀스를선언할때는대괄호안에각요소를쉼표로구분해서넣는다.

public def monthNames = [”January”, ”February”, ”March”,

”April”, ”May”, ”June”,

”July”, ”August”, ”September”,

”October”, ”November”, ”December”];

괄호안의요소가 String이므로이것은 String 시퀀스이다. 다음처럼선언할수도있다. 

public def monthNames: String[] = [ ”January”, .....];

대괄호만 사용하여 빈 문자열을 할당할 수도 있다. 이것은 또한 시퀀스의 디폴트 값이기

도하다. 예를들어, 다음의두선언문은동일하게빈문자열을선언한다.

public var nodes:Node[] = [];

public var nodes:Node[];

시퀀스의 값이 바뀔 때 그 변화에 대한 것을 처리하는 트리거 함수를 할당할 수 있다. 이

에대해서는다음장에서다룬다. 

Integer와 Number의시퀀스를간단하게선언하기위해시작과마지막수를지정하여범위

를 정한다. [1..9]는 1에서 9까지를 나타내는 정수의 시퀀스이고, [1.. < 9]는 9를 포함하

지 않는 1부터 8까지의 정수를 나타낸다. step 함수도 사용할 수 있다. 예를 들어, 짝수만

사용하고 싶다면 [2..100 step 2]가 된다. 수의 시퀀스는 십진수로 소수를 적는다.

[0.1..1.0 step 0.1]과같다. step을정하지않는경우 1이나 1.0이사용된다.

범위는 또한 갈수록 감소하게 정할 수도 있다. 이렇게 하기 위해선 첫 번째 수가 반드시

두 번째 수보다 커야 한다. 그러나 음수의 step 함수가 없으면 빈 시퀀스를 만든다. 이것

은 step의디폴트값이항상양수 1이기때문이다.

var negativeNumbers = [0..–10]; // 빈 시퀀스

var negativeNumbers = [0..–10 step –1]; // 0,–1,–2,...–10

var negativeNumbers = [0..<–10 step –1]; // 0,–1,–2,...,–9 

다른 시퀀스의 요소를 포함하는 시퀀스를 만들 때는 괄호 안에 소스가 되는 시퀀스를 포

함하기만하면된다.

var negativePlusEven = [ negativeNumbers, evenNumbers ];

또한 다른 시퀀스에 Boolean 연산자를 사용하여 새로운 시퀀스를 만들 수 있다. 다른 시

퀀스를 소스로 사용하여 Boolean 연산자를 각 요소에 적용한 다음 조건을 만족하는 것

만을모아새시퀀스를만든다. 다음의예제는양수중 2의 배수만을모아서새로운시퀀

스를만들어낸다. 

var evenIntegers = positiveIntegers[n | n mod 2 == 0];

for루프를 통해서도 시퀀스를 만들 수 있다. for루프의 반복 중‘반환’하는 각각의 객체

를시퀀스에추가한다.

// Text 시퀀스를 생성한다.

var lineNumbers:Text[] = for(n in [1..100]) { 

Text { content: ”{n}” };

};

// indexof 연산자를 사용하여 정수 시퀀스를 만들어낸다.

var indexNumbers = for(n in nodes) { 

indexof n;

};

현재시퀀스의크기를구할때는 sizeof연산자를사용한다.

var numEvenNumbers = sizeof evenNumbers;

시퀀스요소접근

개별적으로요소에접근하려면괄호안에숫자인덱스를사용하면된다.

var firstMonth = monthNames[0];

또한범위를지정하여시퀀스의일부분을취할수있다. 다음의두시퀀스는동일한의미

이다.

var firstQuarter = monthNames[0..2];

var firstQuarter = monthNames[0..<3];

다음의두시퀀스도동일하다. 두번째예에서는시작값만으로범위를지정했고, 이경우

에는인덱스부터다음의모든요소를반환한다.

var fourthQuarter = monthNames[9..11 ];

var fourthQuarter = monthNames[9.. ];

시퀀스를통해반복작업을하려면 for루프를사용한다.

for( month in monthNames) {

println(”{month}”);

}

시퀀스수정

시퀀스에서한요소를수정하는방법은해당인덱스에새로운값을할당하는것이다.

var students = [ ”joe”, ”sally”, ”jim”];

students[0] = ”vijay”;

이 장의 앞에서 보았듯이 JavaFX는 허용적인 언어다. 그래서 현재 시퀀스 범위 밖의 인덱스

에 요소를 할당하면 할당은 무시된다.

이전 예제의 학생 시퀀스를 사용해보자.

students[3] = ”john”;

학생 시퀀스의 현재 크기는 3이고 유효한 가장 큰 인덱스는 2이기 때문에 인덱스 3의 위치에

값을 할당한 것은 수행되지 않는다. 이와 같은 이유로 인덱스 −1 위치에 값을 할당한 것도 시

퀀스의 범위에서 벗어나므로 수행되지 않는다.

게다가 존재하는 시퀀스의 범위에서 벗어난 인덱스에 접근하면 디폴트 값이 반환된다. Number

의 경우에는 0이, String의 경우에는 빈 문자열이, Object의 경우에는 null이 반환된다.

시퀀스에요소를삽입하는방법은 insert 문을사용하는것이다.

// ”vijay” 를 students 시퀀스 마지막에 추가한다.

개발자 노트 DEVELOPER NOTE

insert “vijay“ into students;

// ”mike” 를 students 시퀀스 제일 앞에 삽입한다.

insert “mike“ before students[0];

// ”george” 를 students 시퀀스 두 번째에 삽입한다.

insert “george“ after students[1];

요소를삭제하는방법은 delete문을사용하는것이다.

delete students[0]; // student의 첫 요소를 삭제한다.

delete students[0..1]; // students의 두 번째 요소를 삭제한다.

delete students[0..<2]; // students의 제일 앞에 있는 두 요소를 삭제한다.

delete students[1..]; // student의 두 번째부터 모든 요소를 삭제한다.

delete “vijay“ from students;

delete students; // students의 모든 요소를 삭제한다.

네이티브배열

네이티브 배열native array은 Java 배열을 생성하는 기능이다. 이 기능은 주로 JavaFX와 Java

의배열들을앞뒤로옮길때사용한다. 다음코드는 Java int[]배열을생성하는예제이다.

var ints: nativearray of Integer = [1,2,3] as nativearray of Integer;

네이티브 배열은 시퀀스와 비슷하게 보이지만 시퀀스와 같지는 않다. insert, delete, slice

같은 시퀀스 연산자를 사용할 수 없지만, 다음의 코드처럼 배열의 요소 값을 할당할 수

있다.

ints[2] = 4;

그러나현재배열의범위밖에값을할당하면 ArrayIndexOutOfBounds예외처리가발생한다.

또한네이티브배열에서요소들만큼반복하기위해 for연산자를사용할수있다.

for(i in ints) {

println(i);

}

for(i in ints where i mod 2 == 0) {

println(i);

}

함수

함수는 행동을 정의한다. 함수는 입력 값과 함수 인수를 처리하는 부분을 캡슐화하고 표

현식으로 결과를 반환한다. 변수처럼 함수도 스크립트 함수이거나 인스턴스 함수이다.

스크립트함수는스크립트레벨에서작동하고, 스크립트레벨에서정의한함수나변수에

접근한다. 인스턴스 함수는 객체의 동작을 정의하고, 함수를 선언한 클래스 안에 포함된

다른 인스턴스 함수와 인스턴스 변수에 접근한다. 게다가 인스턴스 함수는 함수가 담긴

스크립트파일안에포함된어떤스크립트레벨의함수와변수에도접근이가능하다.

함수를 선언할 때 function 키워드와 함수명 앞에 public, protected, package 접근 권한자

를 사용할 수 있다. 접근 권한자를 지정하지 않으면 그 함수는 스크립트 파일 안에서만

사용할 수 있다. 모든 함수 인수는 괄호 안에 적는다. 함수의 반환 결과값의 타입도 지정

한다. 반환값의 타입을 생략하면 함수 블록의 마지막 표현식에서 추론한다. Void라는 특

수한반환타입은함수가아무것도반환하지않는다는것을의미한다.

다음 예제의 두 함수 선언은 동일하다. 첫 번째 함수는 함수 블록에서 마지막 표현식이

Glow객체를 위한 객체 상수이기 때문에 반환 타입을 Glow로 추론한다. 두 번째 함수는 명

백하게 Glow라고반환타입을선언하고 return키워드를사용하 다

public function glow(level: Number) {

// 반환 타입을 Glow로 추론

Glow { level: level };

}

public function glow(): Glow { // 반환 타입 명시

return glow(3.0); // return 키워드 명시

}

return키워드는 함수 블록의 마지막 표현식으로 쓸 수 있다. 그러나 if/else나 루프에서

즉시값을반환하고싶다면명시적으로 return을사용해야한다.

JavaFX에서 함수는 함수 안에서 객체이고 변수에 할당할 수 있다. 예를 들어, 함수 변수

를선언하기위해변수에함수를할당하고변수를통해함수를불러온다.

var glowFunction : function(level:Number):Glow;

glowFunction = glow;

glowFunction(1.0);

함수정의는자세히기술하지않을수있다. 예를들어함수변수에,

var glowFunction:function(level:Number): Glow =

function(level:Number) {

Glow { level: level };

};

또는객체상수선언안에서,

TextBox {

columns: 20

action: function() {

println(”TextBox action”);

}

}

수퍼클래스로부터함수를오버라이드할때는 override키워드를사용한다.

class MyClass {

public function print() { println(”MyClass”); }

}

class MySubClass extends MyClass {

override function print() { println(”MySubClass”); }

}

스트링

스트링상수

스트링상수는큰따옴표나작은따옴표를사용하여기술한다. 이렇게두가지를사용하는

이유는 스트링 상수에서 특수 문자를 피하기 위해서이다. 예를 들어, 스트링 상수가 큰따

옴표를 포함하는 경우이다. 스트링을 작은따옴표로 묶음으로써 큰따옴표의 기능을 배제

할수있다. 다음의두예제를보자.

var quote = “Winston Churchill said:

\“Never in the field of human conflict was

so much owed by so many to so few.\““

var quote = ‘Winston Churchill said:

“Never in the field of human conflict was

so much owed by so many to so few.“‘

중괄호를사용하여스트링상수안에표현식을포함시킬수있다.

var name = “Jim“;

// 출력 결과 My name is Jim

println ( “My name is {name}“ );

포함된 표현식은 JavaFX나 Java 표현식이고 객체를 반환한다. 이 객체는 toString()메소

드를사용하여스트링으로변환될것이다

println ( “Today is {java.util.Date{}}“ );

var state = “The state is {if(running) “Running“ else “Stopped“}“;

println(“ The state is {getStateStr()}“ );

println(“The state is {if(checkRunning()) “Running“ else “Stopped“}“);

또한, 문자열상수는여러줄에걸쳐기술할수있다. 

var quote = “Winston Churchill said: “

“\“Never in the field of human conflict was so much owed “

“by so many to so few.\““;

이예제에서스트링들은하나의스트링으로합쳐진다. 따옴표안에서는스트링상수만이

사용되며따옴표밖의공백은무시한다.

유니코드문자는 \u + 숫자 4개의 유니코드를사용하여스트링상수안에입력할수있다.

var thanks = “dank\u00eb“; // dankë

포매팅

스트링상수에포함된표현식은표시방법을기술한포매팅formatting코드를포함할수있다.

var totalCountMessage = ”The total count is {total}”;

total이 정수integer라면 결과 스트링은 10진수로 보일 것이고, total이 수Number라면 결과

스트링은해당지역에맞는포맷으로수를보여줄것이다.



60 3장 JavaFX 첫걸음

var total = 1000.0;

이결과는다음과같다.

The total count is 1000.0

표현식의포맷을정하려면표현식안에포맷코드를넣어야한다. %다음에포맷코드를적

는다. 포맷 코드는 java.util.Formatter클래스에 정의되어 있다. 더 자세히 알아보고 싶다

면 JavaDoc 페이지를참조하면된다(http://java.sun.com/javase/6/docs/api/index.html). 

println(“Total is {%f total}“);     // Total is 1000.000000

println(“Total is {%.2f total}“);   // Total is 1000.00

println(“Total is {%5.0f total}“);  // Total is 1000

println(“Total is {%+5.0f total}“); // Total is +1000

println(“Total is {%,5.0f total}“); // Total is 1,000

스트링에 % 문자열을 포함하려면 %를 덧붙여서 기능을 배제해야 한다(%%). 

println(”%%{percentage}”); // prints %25

국제화

스트링을 국제화internation하는 방법은 스트링 선언 안에“번역 키Translate Key”구문을 사용하

는 것이다. 번역 키를 생성할 때는 스트링에 값을 넣을 때 ##(샵샵)으로 시작하게 하여 지

역에 맞는 언어로 번역되도록 한다. ##은 큰따옴표나 작은따옴표로 시작하기 전에 적는

다. 때로는 키가 []안에서 작성될 수도 있다. 만약 키를 기술하지 않으면 스트링 자체를

키로삼는다.

var postalCode = ## “Zip Code: “;

var postalCode = ##[postal]“Zip Code: “;

앞의 예에서 첫 번째 형태에서의 키는 “zip Code: “이고, 두 번째 형태에서의 키는

“postal“이다

기본적으로로컬라이저localizer는각각의스크립트이름에맞는속성을찾는다. 패키지이름

에스크립트파일이름을더하고, 지역과 .fxproperties파일타입을덧붙인다. 스크립트이

개발자 노트 DEVELOPER NOTE

름이 com.mycompany.MyClass라면 로컬라이저 코드는 com/mycompany/MyClass_xx.fxproperties

를 클래스경로에서 찾을 것이다. xx는 지역정보이다. 예를 들어, 국United Kingdom의 어
English라면 속성 파일 이름은 com/mycompany/MyClass_en_GB.fxproperties가 된다. 프랑스어를

쓰는 캐나다인이라면 com/mycompany/MyClass_fr_CA.fxproperties가 된다. 디폴트 언어가

어라면 속성 파일은 MyClass_en.fxproperties가 된다. 더 자세히 기술한 파일을 우선시하고

적게 기술한 파일은 다음에 찾는다. 예를 들면, 키를 찾을 때 MyClass_en_GB.fxproperties를

먼저 찾고, 찾을 수 없으면 MyClass_en.xproperties를 찾는다. 키를 전혀 찾을 수 없으면 디

폴트로문자열자체를키로삼는다. 다음의예들을보자.

예제 #1 E X A M P L E

println(##”Thank you”);

French – MyClass_fr.fxproperties:

”Thank you” = ”Merci”

German – MyClass_de.fxproperties:

”Thank you” = ”Danke”

Japanese – MyClass_ja.fxproperties:

”Thank you” = ”Arigato”

예제 #2 E X A M P L E

println(##[ThankKey] ”Thank you”);

French – MyClass_fr.fxproperties:

”ThankKey” = ”Merci”

German – MyClass_de.fxproperties:

”ThankKey” = ”Danke”

Japanese – MyClass_ja.fxproperties:

”ThankKey” = ”Arigato”

문자열안에표현식이있을경우상수키는표현식의위치에 %s를포함한다.

(계속)

println(##”Hello, my name is {firstname}”);

이 경우에 키는 ”Hello, my name is %s”가 된다. 마찬가지로 하나 이상의 표현식을 사용

한다면키는각각의표현식위치에 %s를갖는다.

println(##”Hello, my name is {firstname} {lastname}”);

여기서키는 ”Hello, my name is %s %s”가된다.

이파라미터대체는번역문자열에도사용된다.

French – MyClass_fr.fxproperties:

”Hello, my name is %s %s” = ”Bonjour, je m’appelle %s %s”

또다른속성파일을스크립트에적용할수있다. javafx.util.StringLocalizer를사용한다.

StringLocalizer.associate(”com.mycompany.resources.MyResources”, ”com.mycompany”);

com.mycompany 패키지의 스크립트 번역 색인은 디폴트인 스크립트 이름을 쓰는 대신에

com/mycompany/resources/MyResources_xx.fxproperties 속성 파일을 찾는다. 아까처럼 xx는

지역정보를나타내는코드이다.

표현식과 연산자

블록표현식

블록 표현식block expression은 변수 선언 등의 구문과 다른 표현식들을 중괄호 안에 가지고

있는 것이다. 마지막 구문이 표현식이라면 블록 표현식의 값은 마지막 표현식의 값이 되

고, 그렇지않다면블록표현식은값을갖지않는다. 리스트 3.6에는두개의블록표현식

이 있다. 첫 표현식은 subtotal값을 가진다. 두 번째 블록 표현식은 마지막 표현식이 Void

로선언된 println()함수이므로어떤값도가지지않는다.

리스트 3.6 블록 표현식

// 값이 있는 블록 표현식

var total = {

var subtotal = 0;

var ndx = 0;

while(ndx < 100) {

subtotal += ndx;

ndx++;

};

subtotal; // 마지막 표현식

};

// 값이 없는 블록 표현식

{

var total = 0;

var ndx = 0;

while(ndx < 100) {

total += ndx;

ndx++;

};

println(”Total is {total}”);

}

예외처리

throw 문은 Java와같고, java.lang.Throwable로확장하는클래스만을던진다. 

try/catch/finally표현식은 Java와같지만 JavaFX 문법을사용한다.

try {

} catch (e:SomeException) {

} finally {

}

연산자

표 3.2는 JavaFX에서 사용하는 연산자operator의 목록이다. 위의 행부터 순서대로 높은 우

선순위를 가진다. 같은 우선순위를 가진 연산자는 동등하게 계산된다. 할당연산자(=)는

오른쪽에서왼쪽으로계산하고다른모든연산자들은왼쪽에서오른쪽으로계산한다. 괄

호는기본적인계산순서를바꾸는데사용한다. 

조건표현식

if/else

if는 다른 언어에서 정의하는 if와 비슷하다. 조건을 평가하고 참이라면 표현식 블록을

계산한다. 참이아닌경우에 else표현식블록이있다면그표현식블록을계산한다.

if (date == today) {

println(“Date is today“);

}else {

println(“Out of date!!!“);

}

If/else의중요한특징하나는각표현식블록이계산한결과를변수로할당할수있다는

것이다. 

표 3.2 연산자

우선순위 연산자 내용

1 ++/−−(뒤에 붙는 경우) 할당 후에 증가/감소 적용

2 ++/−−(앞에 붙는 경우) 할당 전에 증가/감소 적용

− 마이너스 부호

not 논리 연산자 complement(Boolean 값을 반대로)

sizeof 시퀀스의 크기

reverse 시퀀스의 순서를 반대로

indexof 시퀀스 요소의 인덱스

3 /, ∗, mod 산술 연산자

4 +,  − 산수 연산자

5 ==, != 비교 연산자(Java의 isEquals()와 비슷)

<, <=, >, >= 숫자 비교 연산자

6 instanceof, as 타입 연산자

7 and 논리 연산자 AND

8 or 논리 연산자 OR

9 +=, −=, ∗=, /= 복합 할당

10 =>, tween 애니메이션 보간 연산자

11 = 할당 연산자

var outOfDateMessage = if(date==today) “Date is today“

else “Out of Date“;

또한 표현식 블록은 단순한 표현식보다 복잡해질 수 있다. 리스트 3.7은 if/else 문을 사

용하여 outOfDateMessage에값을할당하는복잡한방식을보여준다.

리스트 3.7 if/else 표현식을 사용한 복잡한 할당

var outOfDateMessage = if(date==today) {

var total = 0;

for(item in items) {

total += items.price;

}

totalPrice += total;

“Date is today“;

} else {

errorFlag = true;

“Out of Date“;

};

앞의 예제에서 블록의 마지막 표현식인 문자열 상수 에러 메시지는 변수로 할당되는 객

체이다. 이것은수를포함하여어떤 JavaFX 객체이건가능하다.

왜냐하면 if/else는표현식블록이라서다른 if/else 문과쓰일수있기때문이다.

var taxBracket = if(income < 8025.0) 0.10

else if (income < 32550.0) 0.15

else if (income < 78850.0) 0.25

else if (income < 164550.0) 0.28

else 0.33;

루프표현식

for

for 문은시퀀스를사용하고, 시퀀스의멤버들에대해반복할수있다. 

(계속)

var daysOfWeek : String[] = [ ”Sunday”, ”Monday”, ”Tuesday” ];

for(day in daysOfWeek) {

println(”{indexof day}). {day}”);

}

수를검사하며반복하는전형적인 for 문처럼대괄호안에시퀀스범위를지정하는수를

사용할수도있다. 

for( i in [0..100]} {

for표현식은 새로운 시퀀스를 반환할 수도 있다. 각각의 반복작업 중 표현식 블록 계산

결과가 객체라면 그 객체는 for 표현식에서 반환할 새로운 시퀀스에 삽입된다. 예를 들

어, 다음의 for표현식 예제에서 한 주의 요일마다 반복하며 새 Text노드가 생성된다. for

표현식은 최종적으로 한 주의 각 요일에 맞는 Text그래픽 요소를 포함하는 새로운 시퀀

스를반환한다.

var textNodes: Text[] = for( day in daysOfWeek) {

Text {content: day };

}

For표현식의다른특징은다른루프를포함할수있다는것이다. 리스트 3.8은중첩루프

를사용한예제를보여준다.

리스트 3.8 중첩 루프

class Course {

var title: String;

var students: String[];

}

var courses = [

Course {

title: ”Geometry I”

students: [ ”Clarke, ”Connors”, ”Bruno” ]

},

Course {

title: ”Geometry II”

students: [ ”Clarke, ”Connors”, ]



표현식과 연산자 67

},

Course {

title: ”Algebra I”

students: [ ”Connors”, ”Bruno” ]

},

];

for(course in courses, student in course.students) {

println(”Student: {student} is in course {course}”);

}

이결과는다음과같다.

Student: Clarke is in course Geometry I

Student: Connors is in course Geometry I

Student: Bruno is in course Geometry I

Student: Clarke is in course Geometry II

Student: Connors is in course Geometry II

Student: Connors is in course Algebra I

Student: Bruno is in course Algebra I

추가 루프는 이전의 루프와 쉼표로 구별한다. 그리고 이전 루프의 어떤 요소라도 참조할

수있다.

또한 where는 where절에서참이라고평가하는요소들만시퀀스에서나오도록할수있다.

var evenNumbers = for( i in [0..1000] where i mod 2 == 0 ) i;

while

while루프는다른언어에서정의하는 while과비슷하다.

var ndx = 0;

while ( ndx < 100) {

println(”{ndx}”);

ndx++;

}

JavaFX의 for루프와 달리 while루프는 어떤 표현식도 반환하지 않는다. 그래서 시퀀스

를생성하는데에사용할수없다는것을기억해야한다.

break/continue

break와 continue는 루프 반복을 제어한다. break는 루프를 완전히 종료할 때 사용한다. 이

것은 해당 지점에서부터 루프 전부를 중지시킨다. 반면에 continue는 단순히 현재 반복을

중지하고, 해당루프는다음반복부터다시한다. 리스트 3.9를보자.

리스트 3.9 break/continue

for(student in students) {

if(student.name == ”Jim”) {

foundStudent = student;

break; // 루프 모두 중지

// 더는 student를 검사하지 않는다.

}

}

for(book in Books ) {

if(book.publisher == ”Addison Wesley”) {

insert book into bookList;

continue; // 다음 book을 검사하러 간다.

}

insert book into otherBookList;

otherPrice += book.price;

}

타입 연산자

instanceof연산자는 객체의 클래스 타입을 검사한다. 반면에 as연산자는 객체를 다른 클

래스로 캐스팅한다. as는 몇몇 특수한 클래스의 함수를 사용하기 위해서 보통의 클래스

를 특수 클래스로 캐스팅할 때에 유용하다. 물론 객체는 그 클래스에서 상속받았어야 하

고, 이를 검사할 때 instanceof가 유용하다. 그렇지 않은 객체를 캐스팅하려 하면 예외가

발생할것이다.

다음의 리스트에서 printLower() 함수는 문자열을 소문자로 바꾼다. 그러나 다른 타입의

객체는 그대로 출력한다. 먼저 객체가 문자열인지를 테스트하고 맞다면 객체는 as 연산

자를사용하여문자열로캐스팅한다. 그다음문자열의 toLowerCase()메소드가결과를소

문자로변환한다. 리스트 3.10은 instanceof와 as 연산자를사용한예이다.  

리스트 3.10 타입 연산자

function printLower(object: Object ) {

if(object instanceof String) {

var str = object as String;

println(str.toLowerCase());

}else {

println(object);

}

}

printLower(”Rich Internet Application”);

printLower(3.14);

커맨드라인인수접근

익스포트된클래스, 변수, 함수를선언하지않는순수한스크립트의커맨드라인인수command-

line arguments는 javafx.lang.FX.getArguments():String[]함수를사용하여얻을수있다. 이함수

는스크립트가시작할때커맨드라인에서스크립트에넘긴인수를가지고있는문자열시

퀀스를 반환한다. 이 함수의 다른 버전인 javafx.lang.FX.getArguments(key:String):String[]

함수는애플릿처럼다른방식으로사용할때에이름-값의쌍으로인수를전달한다. 비슷한

방식으로시스템속성을얻는 javafx.lang.FX.getProperty(key:String):String[]함수도있다.

만약 스크립트가 익스포트된 클래스, 변수, 함수를 가지고 있다면 스크립트 레벨에서 정

의한 run함수를통해서인수를얻는다.

public function run(args:String[] ) {

for(arg in args) {

println(”{arg}”);

}

}

익스포트된 멤버를 사용한 loose 표현식

스크립트 레벨의(클래스 선언 내부가 아닌) 변수, 함수, 표현식을 loose 표현식이라고 한다.

이 변수와 함수가 스크립트에서 private일 때, 커맨드라인에서 스크립트를 실행한다면 특정

개발자 노트 DEVELOPER NOTE

run 함수는 필요하지 않다. 그러나 이 표현식 중 어느 하나라도 public, public-read,

protected, package 접근 권한으로 스크립트의 외부에 익스포트되면, 스크립트를 직접 실행

할 때 run 함수가 필요하다. 이 run 메소드는 익스포트된 변수와 함수를 캡슐화한다.

빌트인함수와변수

모든 JavaFX 스크립트에서 선언 및 정의 없이 사용 가능한 함수들이 있다. 이 함수들은

javafx.lang.Builtins에정의되어있다.

이 중 하나인 println()을 이미 보았다. println()은 객체 인수를 받아서 한 번에 한 줄씩

콘솔에 출력한다. 이것은 Java 메소드인 System.out.println()과 비슷하다. 이것과 비슷한

함수인 print()는 새로운 줄 없이 인수를 출력한다. 인수의 toString()메소드는 문자열을

출력하기위해호출된다. 

println(“This is printed on a single line“);

print(“This is printed without a new line“);

javafx.lang.Builtins의또다른함수중 isInitialized()가있다. 이메소드는 JavaFX 객체

를 받아서 그 객체가 완전히 초기화되었는지를 알려준다. 변수 트리거에서 초기화 중인

객체의 현재 상태를 정할 때 유용하다. 객체를 초기화하도록 전달한 후 기능을 실행하기

전에 약간의 시간이 필요하다. 예를 들어, 리스트 3.11에서 isInitialized()가 replace 트

리거에서쓰이는것을보자.

리스트 3.11 isInitialized()

public class Test {

public var status: Number on replace {

// status에 값이 할당될 때까지 초기화되지 않는다.

if(isInitialized(status)) {

commenceTest(status);

}

}

public function commenceTest(status:Number) : Void {

println(“commenceTest status = {status}:);

}

}

이예제에서는 Test클래스가처음초기화될때인스턴스변수 status는먼저디폴트값인

0.0을가진다. 그리고그다음에 on replace표현식블록이계산된다. 그러나이는 status를

uninitial-ized상태로둔다. 값이 status에대입이될때만상태는 initialized로변경될것이

다. 다음을보자.

var test = Test{}; // status는 uninitialized

test.status = 1;   // 이제 status는 initialized가 된다.

이 경우 Test가 Test{} 객체 상수를 사용하여 만들어지면 status는 디폴트 값인 0.0을 가

진다. 그러나 initialized가 아니므로 commenceTest는 객체를 생성하는 동안 호출되지 않

을 것이다. 이제 status에 값을 대입하면 상태는 initialized로 변경되고 commenceTest도

호출된다. status에 디폴트 값을 대입하면 비록 그 값이 0일지라도 status는 즉시

initialized가된다는것을주의하자. 다음의예제를보자.

public class Test {

public var status: Number = 0 on replace {

// 즉시 initialized가 된다.

if(isInitialized(status)) {

commenceTest(status);

}

}

}

마지막으로 다룰 빌트인 함수는 isSameObject()이다. isSameObject()는 두 인수가 실제로

같은인스턴스인지를판단한다. 이함수는 == 연산자와다르다. JavaFX에서 == 연산자는

두 개의 객체가 같은 값인지를 보지만 그것이 같은 인스턴스임을 의미하지 않는다. ==

연산자는 Java 함수 isEquals()와 비슷하고 JavaFX의 isSameObject는 Java의 == 연산자와

비슷하다. Java를알고있다면조금헷갈릴것이다.

빌트인 변수에는 __DIR__와 __FILE__이 있다. __FILE__는 JavaFX 클래스를 리소스 URL 문

자열로 가지고 있다. __DIR__은 현재의 클래스를 담고 있는 디렉터리에 해당하는 리소스

URL 문자열을가지고있다. 

println(“DIR = {__DIR__}“);

println(“FILE = {__FILE__}“);

// 이미지의 위치 입력

var image = Image { url: “{__DIR__}images/foo.jpeg“ };

다음의예는디렉터리클래스경로와 JAR 기반클래스경로를사용한출력의비교이다. 

클래스경로에서 Jar 파일을 사용한다.

$javafx -cp Misc.jar misc.Test

DIR = jar:file:/export/home/jclarke/Documents/Book/FX/code/Chapter3/Misc/dist/Misc.jar!/misc/

FILE = jar:file:/export/home/jclarke/Documents/

Book/FX/code/Chapter3/Misc/dist/Misc.jar!/misc/Test.class

디렉터리 클래스경로 사용

$ javafx -cp . misc.Test

DIR = file:/export/home/jclarke/Documents/Book/FX/code/Chapter3/Misc/dist/tmp/misc/

FILE = file:/export/home/jclarke/Documents/Book/FX/code/Chapter3/Misc/dist/tmp/misc/Test.class

__DIR__의 뒤에 오는 슬래시(/)를 주의하라

슬래시는 이미 __DIR__에 포함되어 있으므로 이미지와 같은 리소스에 대한 경로를 만들 때 슬

래시를 추가하지 않아야 한다. Image{ url: “{__DIR__}image/foo.jpeg“}은 옳다. 

Image{ url: “{__DIR__}/image/foo.jpeg“}은 틀리다. __DIR__ 뒤에 슬래시를 추가하면 이미지

를 찾을 수 없고, 왜 안 되는지를 알아내기 위해 머리를 쥐어짜야 할 것이다.

요약

이 장에서는 JavaFX 스크립트 언어의 핵심 개념을 다루었다. 스크립트를 구성하는 요소

와 클래스를 구성하는 요소들도 보았다. 스크립트와 인스턴스 변수를 선언하는 방법과

시퀀스를만들고수정하는방법, 그리고논리흐름을제어하는방법도보았다.

이제 JavaFX 스크립트 언어의 문법과 연산자의 기본을 이해하게 되었다. 이제 이것을 사

용할 차례이다. 뒤의 장들에서는 JavaFX의 핵심 기능들을 반복하여 연습하고 그 기능들

을 이용하기 위해 JavaFX 스크립트 언어를 사용하는 방법을 다룰 것이다. 다음 장에서는

JavaFX 런타임에서의데이터동기화지원을알아보자.

개발자 주의사항 DEVELOPER WARNING


